/* READ COMMENTS:

This macro applies the polytomous logistic regression approach (QPL), which models the offspring genotypes, conditional on the parental genotypes and the offspring's QT value.  This macro assumes that the trios are all complete and that the marker under study is either di-allelic or has alleles that have been meaningfully grouped into two categories.  If some trios are incomplete, use the other macro provided in the same website. The hypothesis testing results will not depend on which of the two alleles under study is designated as the "variant".  The method is described in "A method for using complete and incomplete trios to localize genes related to a quantitative trait" by Kistner and Weinberg, currently under review.  The QT need not be normally distributed, and HWE is not required. The stratification on parental genotypes confers robustness against hidden population stratification and admixture. 

Should be used with SAS Version 8 or higher. Runs in batch. 

This code works for the following SAS data.  

Each record must consist of a M,F,C, and X, where 

M= # of copies of the variant allele that the mother carries,

F= # of copies of the variant allele that the father carries,

C= # of copies of the variant allele that the offspring carries, and

X= the quantitative trait, as measured for the offspring.

YOU MUST ENTER THE NAME OF YOUR DATA SET IN THE FIRST %LET STATEMENT BELOW!

*/

%LET YOUR_DATA=ENTERHERE;

ODS LISTING CLOSE;

DATA SAMPLE;                  /* defining mating type variables and the reference category for the model */

  SET &YOUR_DATA;

  RETAIN COUNT_0 0 COUNT_2 0;

  MTYPE01=0;

  MTYPE11=0;

  MTYPE12=0;

  IF (((M=0) AND (F=1)) OR ((M=1) AND (F=0))) THEN MTYPE01=1;

  IF ((M=1) AND (F=1)) THEN MTYPE11=1;

  IF (((M=1) AND (F=2)) OR ((M=2) AND (F=1))) THEN MTYPE12=1;

  IF (C=1) THEN ORDER=1;

           ELSE ORDER=0;

  IF (MTYPE01) OR (MTYPE11) OR (MTYPE12) THEN OUTPUT;

RUN;

/* Sorting data so that the reference cell is C=1. */

PROC SORT DATA=SAMPLE OUT=PLOGDATA;

BY ORDER;

RUN;

ODS TRACE OFF;

ODS OUTPUT FitStatistics=Fstats1 ParameterEstimates=ParmEsts;

PROC LOGISTIC DATA=PLOGDATA ORDER=DATA;

MODEL C = X MTYPE01 MTYPE11 MTYPE12 / NOINT LINK=GLOGIT;

CONTRAST "OVERALL" X 1;

RUN;

ODS TRACE OFF;

ODS OUTPUT FITSTATISTICS=FSTATS2;

PROC LOGISTIC DATA=PLOGDATA ORDER=DATA;

MODEL C = MTYPE01 MTYPE11 MTYPE12 / NOINT LINK=GLOGIT;

RUN;

/* Computing the test statistic */

DATA LR;

  MERGE FSTATS1(WHERE=(CRITERION="-2 Log L") RENAME=(WithCovariates=WX)) 

        FSTATS2(WHERE=(CRITERION="-2 Log L") RENAME=(WithCovariates=WOX));

  DF=2;

  LRT=WOX-WX;

  PROBCHISQ=1-PROBCHI(LRT,DF);

  DROP WITHOUTCOVARIATES CRITERION WX WOX;

run;

ODS LISTING;

TITLE1 "Test of Linkage and Association";

PROC PRINT DATA=LR;

RUN;

TITLE1 "Estimates of Parameters from Model";

PROC PRINT DATA=PARMESTS;

RUN;

